
Vocal simulation 
 
This is an example of the use of the FOF generator to simulate a vocal sound. It is based 
on the generation of a sound with 5 formant regions, set in the variables iform1-5: 
 
iform1  = 800                  ; 5 formant regions 
iform2  = 1150 
iform3  = 2800 
iform4  = 3500 
iform5  = 4950 
 
These formant regions have each 5 different amplitude levels, set by iaf1-5: 
 
iaf1     = ampdb(0)                 ; formant amplitudes 
iaf2     = ampdb(-4)                   
iaf3     = ampdb(-20) 
iaf4     = ampdb(-36) 
iaf5     = ampdb(-60) 
 
Each formant region has a different bandwidth, set in ibw1-5: 
 
ibw1   = 50 
ibw2   = 60 
ibw3   = 170 
ibw4   = 180 
ibw5   = 200 
 
These parameters are set for a particular vowel on a particular type of voice. In this case, 
the values refer to an ‘a’ sung by an alto voice. 
 
In this instrument, we will make the sound gliss from one note to another,  given by ifun1 
and ifun2. We will also make the octaviation parameter vary from ioct1 to ioct2. 
 
koct    linseg     ioct1, p3*.3, ioct1, p3*.4, ioct2, p3*.3, ioct2 
kfun    expseg     ifun1, p3*.3, ifun1, p3*.4, ifun2, p3*.3, ifun2 
 
For a successful vocal simulation is important to include some vibrato and a little 
frequency ‘jitter’. The vibrato intensity (width) will also increase as we go towards 
the second note. The vibrato change will follow the change in the fundamental. Jitter is 
created by a random sig generator and the vibrato is a simple sine oscillator with an 
envelope: 
 
ivib1     =   .5*log(ifun1)       ; vibrato intensity 
ivib2     =   log(ifun2) 
 
kvib    expseg     ivib1, p3*.3, ivib1, p3*.4, ivib2, p3*.3, ivib2 
 
aj     randi  kvib, 125 
kv     linen  kvib, .6, p3,.2 
av     oscil  kv, 5.8, 1 

 
The resulting sound from the 5 FOF generators is mixed, scaled and shaped by an 
exponential envelope (also sent to a global reverb): 
 



kenv envlpx  iamp, .05, isecs, .3, irshape,  .9, .01 ; overall envelope 
asum   =     a1+a2+a3+a4+a5                          ; mix all formant regions 
garev = asum*iscal*kenv      
      out   garev ; output 
 
Here’s the complete orchestra and score files: 
 
; Formant synthesis example 
; Victor lazzarini, 1998 
sr = 44100 
kr = 4410 
ksmps = 10 
nchnls = 1 
garev init 0 
 
 
instr 1 
 
iol     = 30                   ; max overlaps 
ifun1    = p5                  ; fundamental 1 
ifun2    = p6                  ; fundamental 2 
iamp     = p4*.6                  ; overall amp 
isecs     = p3                 ; note duration in secs 
irshape  = 3                   ; envlpx rise shape 
isine    = 1                   ; sinewave 
itxshape = 2                   ; FOF granules tex shape 
 
iform1  = 800                  ; 5 formant regions 
iform2  = 1150 
iform3  = 2800 
iform4  = 3500 
iform5  = 4950 
 
iaf1     = ampdb(0)                 ; formant amplitudes 
iaf2     = ampdb(-4)                   
iaf3     = ampdb(-20) 
iaf4     = ampdb(-36) 
iaf5     = ampdb(-60) 
iscal   = 1/(iaf1+iaf2+iaf3+iaf4+iaf5) ; scale output 
 
ibw1   = 50 
ibw2   = 60 
ibw3   = 170 
ibw4   = 180 
ibw5   = 200 
 
ioct1    = p7                   ; FOF parameters 
ioct2    = p8                   ; octaviation 
itx      = p9                   ; temps d'excitation (granule rise time!)                
idur     = .02                  ; burst duration 
iatt     = .007                 ; attenuation 
 
ivib1     =   .5*log(ifun1)       ; vibrato intensity 
ivib2     =   log(ifun2) 
 
; dynamic variation of FOF parameters 
koct    linseg     ioct1, p3*.3, ioct1, p3*.4, ioct2, p3*.3, ioct2 
kfun    expseg     ifun1, p3*.3, ifun1, p3*.4, ifun2, p3*.3, ifun2 
kvib    expseg     ivib1, p3*.3, ivib1, p3*.4, ivib2, p3*.3, ivib2 
 
      ;jitter & vibr (p20) 
aj     randi  kvib, 125 
kv     linen  kvib, .6, p3,.2 
av     oscil  kv, 5.8, 1 
 
      ;fundamental + jitter + vibr 
afun   =      kfun+av+aj 
 
; fof generators 
;        amp fun   fomfr   oct   ban    tex  idur  iatt   olap ifna  ifnb   dur  ph  mode 



a1 fof iaf1, afun, iform1, koct, ibw1, itx, idur, iatt,  iol, isine, itxshape, isecs, 0,1 
a2 fof iaf2, afun, iform2, koct, ibw2, itx, idur, iatt,  iol, isine, itxshape, isecs, 0,1 
a3 fof iaf3, afun, iform3, koct, ibw3, itx, idur, iatt,  iol, isine, itxshape, isecs, 0,1 
a4 fof iaf4, afun, iform4, koct, ibw4, itx, idur, iatt,  iol, isine, itxshape, isecs, 0,1 
a5 fof iaf5, afun, iform5, koct, ibw5, itx, idur, iatt,  iol, isine, itxshape, isecs, 0,1 
 
 
kenv envlpx  iamp, .05, isecs, .3, irshape,  .9, .01 ; overall envelope 
asum   =     a1+a2+a3+a4+a5                          ; mix all formant regions 
garev = asum*iscal*kenv      
      out   garev ; output 
       
endin 
 
instr 10 
 
; global reverb  
 
ibal = p4 
aout  reverb garev, 0.9 
      out  aout*ibal 
 
endin 
 
 
f1 0 4096 10 1 ; sine 
f2 0 1024  19  .5  1  270  1  ; granule tex shape 
f3 0 1025 05 0.01 256 .3 256 .8 512 1 ; envlpx rise shape 
 
;          amp    fun1   fun2   oct1 oct2  tx    
i1  .5  1  12000  330    330    0   0    .003      
i1  +  .5  16000  660    440    0   0    .003 
i1  +  1.5 10000  400    400    0   0    .003  
i1  +  1  15000  660    660    0   0    .003 
i1  +  .3  12000  660    600    0   0    .003 
i1  +  .3  10000  600    600    0   0    .003 
i1  +  .3  9000  510    510    0   0    .003 
i1  +  .3  8000  440    400    0   0    .003 
i1  +  2  10000  345    310    0   0    .003 
i1  +  2  13000  225    220    0   0    .003 
i10 .5 11 .5 
e 

